Algèbre linéaire et géométrie

(201-NYC)

Résolution de problèmes

RP 3

Déroulement Vous devez prendre connaissance du texte intitulé «Session Maple pour l'exercice RP-3». Ce texte vous donne toute l'information pertinente quant à la manière dont est appliquée la méthode de Gauss-Jordan afin de déterminer l'équation d'une courbe lorsqu'on connaît un certain nombre de points de celle-ci.

On se situera dans le contexte où la courbe d'ajustement à trouver est celle d'une conique : parabole, ellipse ou hyperbole.

Le texte dont il est question constitue en lui-même un modèle à suivre quant à la manière de documenter en Maple la zone des résultats ainsi que les développements mathématiques de vos solutions.

Ce texte, que l'on retrouve à la fin du présent document, est présenté sous la forme d'une session Maple.

Objectif — Appliquer la méthode de Gauss-Jordan dans le cadre d'une introduction à la technique d'ajustement analytique (data-fitting).

Plus précisément, prendre connaissance de la manière dont la méthode de Gauss-Jordan apporte une réponse au problème suivant : connaissant un certain nombre de points dans le plan cartésien $(x_1, y_1), (x_2, y_2), \ldots, (x_n, y_n)$, peut-on déterminer les coefficients a, b, c, d, e et f de telle manière que le graphique de la courbe d'équation

$$a\,x^2 + b\,xy + c\,y^2 + d\,x + e\,y + f = 0$$

passe par ces points ?

Énoncé des problèmes

No. 1 a) Faire une recherche sur l'équation générale

$$a\,x^2 + b\,x\,y + c\,y^2 + d\,x + e\,y + f = 0$$

afin de déterminer les conditions sur les coefficients a, b, c, d, e et f pour que l'équation en question soit celle

i) d'une ellipse
ii) d'un cercle
iii) d'une parabole
iv) d'une hyperbole

Formuler les résultats de votre recherche dans une zone de texte sans crochet de bloc.

b) Sur la base des coefficients de chaque équations ci-dessous, déterminer la nature des coniques d'équations

i) $2x^2 + 8xy + 7y^2 + x - 3y + 8 = 0$
ii) $2x^2 + 8xy + 8y^2 + x - 3y + 8 = 0$
iii) $2x^2 + 8xy + 9y^2 + x - 3y + 8 = 0$

No. 2 a) Déterminer l'équation de la conique passant par les points (-4,1), (-1,2), (3,2), (5,1) et (7,-1). À partir de l'équation que vous avez obtenue, de quelle conique s'agit-il ? Justifier votre réponse sur la base des coefficients de votre équation.

b) Superposer, dans un même graphique, le tracé des points donnés ainsi que celui de votre conique. Pour illustrer les points, utiliser la macro-commande `disk` de la bibliothèque `plottools`. Ajuster la valeur du rayon afin que le tracé illustre correctement celui d'un point du plan. Préciser évidemment un même rayon pour tous les points. De plus, afin d'être en mesure de bien mettre en évidence dans un même graphique les caractéristiques demandées, consulter l'aide de la macro-commande `plot_real_curve` de l'extension `algcurves` en tapant dans une zone de requêtes « `?algcurves[plot_real_curve]` ».

No. 3 Soit les trois points suivants : (4,3), (1,2) et (2,0).

a) Déterminer l'équation du cercle passant par ces trois points.

b) Illustrer, dans un même graphique, les trois points et le cercle.

Remarque : Utiliser les résultats de votre recherche afin de vous permettre de poser au départ la forme générale de l'équation d'un cercle.

No. 4 a) Montrer qu'il est possible d'ajuster un infinité de paraboles passant par les trois points du numéro précédent. (Soit une parabole verticale, une parabole horizontale et une infinité de paraboles obliques)

b) Donner les équations de la parabole verticale et de la parabole horizontale. Donner aussi l'équation d'une parabole oblique de votre choix.

c) À l'aide d'une matrice de format 1×3, afficher simultanément les trois tracés suivants :
— la parabole verticale, les trois points et le cercle obtenu au numéro précédent ;
— la parabole horizontale, les trois points et le cercle obtenu au numéro précédent ;
— la parabole oblique de votre choix, les trois points et le cercle obtenu au numéro précédent ;

Session Maple pour les exercices RP-3

Initialisation

```
> restart;
  with(LinearAlgebra,GenerateMatrix,BackwardSubstitute,
                    ReducedRowEchelonForm):
  with(plottools,disk):
  with(plots,display):
  with(algcurves,plot_real_curve):
```

Équation d'une droite

Nous allons déterminer l'équation d'une droite connaissant deux points. Mais, au lieu de le faire de manière habituelle comme on vous l'a montré en géométrie analytique avec un calcul de la pente, nous allons plutôt développer une approche qui pourra être applicable pour des courbes beaucoup moins simple que celle d'une droite.

Déterminons l'équation de la droite passant par les points (1,2) et (3,7).

L'équation générale d'une droite est de la forme

$$a\,x + b\,y + c = 0$$

Le problème consiste donc à trouver les valeurs des coefficients a, b et c afin que les coordonnées de chaque point donné puisse vérifier cette équation. Par exemple, dans le cas du point (1,2), on doit avoir nécessairement $a\,1 + b\,2 + c = 0$. En considérant simultanément les deux points à la fois, il s'agit donc de résoudre le système suivant.

$$\begin{cases} a + 2b + c = 0 \\ 3a + 7b + c = 0 \end{cases}$$

Le système à résoudre est un système homogène. Ce système est clairement un système compatible avec une infinité de solutions (pourquoi ?). Résolvons ce système d'équations par la méthode de Gauss-Jordan.

Saisissons directement la matrice augmentée du système à résoudre.

```
> `A|B`:=<<1|2|1|0>,
          <3|7|1|0>>;
```

$$A|B := \begin{bmatrix} 1 & 2 & 1 & 0 \\ 3 & 7 & 1 & 0 \end{bmatrix}$$

Obtenons la forme de Gauss-Jordan.

```
> `A|B_GJ`:=ReducedRowEchelonForm(A|B`);
```

$$A|B\_GJ := \begin{bmatrix} 1 & 0 & 5 & 0 \\ 0 & 1 & -2 & 0 \end{bmatrix}$$

Nous obtenons $a = -5t$, $b = 2t$ et $c = t$. L'équation générale de la droite est alors $-5t\,x + 2t\,y + t = 0$. En choisissant $t = 1$, nous déduisons l'équation

$$-5x + 2y + 1 = 0$$

Illustrons dans un même graphique les deux points donnés ainsi que la droite d'équation

$$-5x + 2y + 1 = 0$$

```
> Équation_Droite:= y=solve(-5*x+2*y+1=0,y);
```

$$quation_Droite := y = -\frac{1}{2} + \frac{5}{2}\,x$$

```
> P1:=disk([1,2], 0.15, color=orange):
  P2:=disk([3,7], 0.15, color=orange):
  Droite:=plot([x,rhs(Équation_Droite),x=-1..4],color=navy):
  display({P1,P2,Droite},axesfont=[TIMES,ROMAN,10],scaling=constrained);
```




```
>
```

Sections coniques

Soit l'équation quadratique à deux variables

$$a\,x^2 + b\,x\,y + c\,y^2 + d\,x + e\,y + f = 0$$

Si cette équation possède des racines réelles, le tracé de son graphe sera une certaine courbe dans le plan cartésien. Dans le cas où au moins un des coefficients a, b ou c est non nul, le tracé obtenu est appelé une conique.

Déterminons la conique passant par les points (-1,0), (0,1), (2,-1), (2,2) et (0,-3). Comme dans la section précédente, il s'agit de trouver les valeurs des coefficients a, b, c, d, e et f afin que les coordonnées de chaque point donné puisse vérifier cette équation. Par exemple, avec le point (2,-1) découle l'équation

$$a\,2^2 + b\,2\,(-1) + c\,(-1)^2 + d\,2 + e\,(-1) + f = 0 \Longleftrightarrow 4\,a - 2\,b + c + 2\,d - e + f = 0$$

Obtenons le système d'équations à résoudre en considérant les cinq points pris à la fois.

```
> Équation_générale:=a*x^2+b*x*y+c*y^2+d*x+e*y+f=0;
```

$$\mathit{Équation\_générale} := a\,x^2 + b\,x\,y + c\,y^2 + d\,x + e\,y + f = 0$$

```
> Éq1:=eval(Équation_générale,[x=-1,y=0]);
  Éq2:=eval(Équation_générale,[x=0,y=1]);
  Éq3:=eval(Équation_générale,[x=2,y=2]);
  Éq4:=eval(Équation_générale,[x=2,y=-1]);
  Éq5:=eval(Équation_générale,[x=0,y=-3]);
```

$$\mathit{Éq1} := a - d + f = 0$$

$$\mathit{Éq2} := c + e + f = 0$$

$$\mathit{Éq3} := 4\,a + 4\,b + 4\,c + 2\,d + 2\,e + f = 0$$

$$\mathit{Éq4} := 4\,a - 2\,b + c + 2\,d - e + f = 0$$

$$\mathit{Éq5} := 9\,c - 3\,e + f = 0$$

Il s'agit alors de résoudre le système S.

$$\mathrm{S}: \left\{ \begin{array}{rcrcrcrcrcrcl} a & & & & & & -d & & & + & f & = & 0 \\ & & & & c & + & & & e & + & f & = & 0 \\ 4a & + & 4b & + & 4c & + & 2d & + & 2e & + & f & = & 0 \\ 4a & - & 2b & + & c & + & 2d & - & e & + & f & = & 0 \\ & & & & 9c & & & - & 3e & + & f & = & 0 \end{array} \right.$$

Créons la matrice augmentée du système S.

```
> S:=[Éq||(1..5)];     # || est l'opérateur de concaténation
  Variables:=[a,b,c,d,e,f];
  `A|B`:=GenerateMatrix(S,Variables,augmented=true);
```

$$S := [a - d + f = 0,\, c + e + f = 0,\, 4\,a + 4\,b + 4\,c + 2\,d + 2\,e + f = 0,\\ 4\,a - 2\,b + c + 2\,d - e + f = 0,\, 9\,c - 3\,e + f = 0]$$

$$\mathit{Variables} := [a,\, b,\, c,\, d,\, e,\, f]$$

$$A|B := \begin{bmatrix} 1 & 0 & 0 & -1 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 1 & 1 & 0 \\ 4 & 4 & 4 & 2 & 2 & 1 & 0 \\ 4 & -2 & 1 & 2 & -1 & 1 & 0 \\ 0 & 0 & 9 & 0 & -3 & 1 & 0 \end{bmatrix}$$

Obtenons la forme de Gauss-Jordan de la matrice augmentée du système S.

```
> `A|B_GJ`:=ReducedRowEchelonForm(A|B`);
```

$$A|B\_GJ := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & \frac{7}{18} & 0 \\ 0 & 1 & 0 & 0 & 0 & \frac{-1}{2} & 0 \\ 0 & 0 & 1 & 0 & 0 & \frac{1}{3} & 0 \\ 0 & 0 & 0 & 1 & 0 & \frac{-11}{18} & 0 \\ 0 & 0 & 0 & 0 & 1 & \frac{2}{3} & 0 \end{bmatrix}$$

Ce système homogène S possède évidemment une infinité de solutions. Obtenons ces solutions sur la base de la variable libre f.

```
> Racines:=BackwardSubstitute(A|B_rref`);
```

$$Racines := \left[-\frac{7}{18}\_t_1,\ \frac{1}{2}\_t_1,\ -\frac{1}{3}\_t_1,\ \frac{11}{18}\_t_1,\ -\frac{2}{3}\_t_1,\ \_t_1\right]$$

Reformulons la solution paramétrique en terme du paramètre t.

```
> Racines_reformulées:=subs(_t[1]=t,eval(Racines));
```

$$Racines\_reformulées := \left[-\frac{7}{18}t,\ \frac{1}{2}t,\ -\frac{1}{3}t,\ \frac{11}{18}t,\ -\frac{2}{3}t,\ t\right]$$

Documentons dans la zone des résultats l'ensemble solution du système S.

```
> Solution:=zip((x,y)->x=y,Variables,convert(Racines_reformulées,list));
```

$$Solution := \left[a = -\frac{7}{18}t,\ b = \frac{1}{2}t,\ c = -\frac{1}{3}t,\ d = \frac{11}{18}t,\ e = -\frac{2}{3}t,\ f = t\right]$$

On obtient $a = -\frac{7\,t}{18}$, $b = \frac{t}{2}$, $c = -\frac{t}{3}$, $d = \frac{11\,t}{18}$, $e = -\frac{2\,t}{3}$ et $f = t$.

En choisissant $t = 18$ nous éliminerons les fractions.

```
> Solution_entière:=eval(Solution,t=18);
```

$$[a = -7,\ b = 9,\ c = -6,\ d = 11,\ e = -12,\ f = 18]$$

On obtient alors l'équation suivante de la conique passant par ces cinq points.

```
> Équation_de_la_conique:=eval(Équation_générale,Solution_entière);
```

$$Équation\_de\_la\_conique := -7\,x^2 + 9\,x\,y - 6\,y^2 + 11\,x - 12\,y + 18 = 0$$

Avant même de tracer cette conique, nous sommes maintenant en mesure d'identifier la conique passant par ces cinq points. Sur la base des coefficients de l'équation, cette conique est une ellipse. (Pourquoi ?)

Terminons par la superposition, dans un même graphique, des cinq points donnés ainsi que l'ellipse d'équation

$$-7\,x^2 + 9\,x\,y - 6\,y^2 + 11\,x - 12\,y + 18 = 0$$

```
> P1:=disk([-1,0],0.1,color=orange):
  P2:=disk([0,1], 0.1,color=orange):
  P3:=disk([2,2], 0.1,color=orange):
  P4:=disk([2,-1],0.1,color=orange):
  P5:=disk([0,-3],0.1,color=orange):
  Ellipse:=plot_real_curve(lhs(Équation_de_la_conique),x,y):
  display({P||(1..5),Ellipse},axesfont=[TIMES,ROMAN,10],
           scaling=constrained);
```